

# Efficient Parallel Stencil Convolution in Haskell

Ben Lippmeier   Gabriele Keller

School of Computer Science and Engineering  
University of New South Wales, Australia  
{benl, keller}@cse.unsw.edu.au

Simon Peyton Jones

Microsoft Research Ltd  
Cambridge, England  
simonpj@microsoft.com

## Abstract

Stencil convolution is a fundamental building block of many scientific and image processing algorithms. We present a declarative approach to writing such convolutions in Haskell that is both efficient at runtime and implicitly parallel. To achieve this we extend our prior work on the Repa array library with two new features: partitioned and cursored arrays. Combined with careful management of the interaction between GHC and its back-end code generator LLVM, we achieve performance comparable to the standard OpenCV library.

**Categories and Subject Descriptors** D.3.3 [Programming Languages]: Language Constructs and Features—Concurrent programming structures; Polymorphism; Abstract data types

**General Terms** Languages, Performance

**Keywords** Arrays, Data parallelism, Haskell

## 1. Introduction

This paper addresses an implicit challenge put to us by Rami Mukhtar of NICTA (the Australian equivalent of INRIA). At the time, Rami was starting a project on writing image processing algorithms in declarative languages. Having read our previous work on the Repa library for parallel arrays [13], he took it to heart, and promptly implemented the Canny edge detection algorithm [6] as a benchmark. Unfortunately, he then informed us that the Repa version was more than 10x slower than the equivalent implementation in OpenCV [5], an industry standard library of computer vision algorithms. Due to this, he instead based his project around the Accelerate EDSL [8] for writing parallel array codes on GPGPUs, produced by a competing (but friendly) faction in our research group. Clearly, we could not let this stand.

Simply put, our aim is to implement parallel image processing algorithms that run as fast (faster!) than the highly optimised ones for imperative languages. We also want to write this code directly in Haskell and use the GHC runtime system, instead of, say, implementing an EDSL that produces LLVM or CUDA code. Using Haskell directly gives us access to GHC’s powerful inliner and simplifier, which we use to convert declarative code into the tight loops we rely on for performance. The GHC runtime provides the primitives we use to implement parallelism in a portable way.

At the core of many image processing algorithms is the 2D convolution operator  $*$ , whose definition is as follows:

$$(A * K)(x, y) = \sum_i \sum_j A(x + i, y + j) K(i, j)$$

Here,  $A$  is the image being processed, and  $K$  is the *convolution kernel* or *stencil*. The *stencil* is a small matrix, with typical dimensions 3x3 or 1x5, that defines a transformation on the image. Typical transformations include the Gaussian blur, and the Sobel differentiation operator, both of which are used in the Canny edge detection algorithm. For this paper we focus on the efficient parallel implementation of stencil convolution, though we will return to the larger Canny algorithm near the end. As we are primarily interested in image processing we also focus on arrays of rank 2, though our techniques are equally applicable to arrays of higher rank.

Our contributions are as follows:

- An array fusion approach to writing stencil functions in Haskell that yields performance comparable to the industry standard OpenCV library.
- To achieve this we extend our previous approach [13] with two new features: partitioned and cursored arrays. These allow us to optimise array programs that use different functions to construct the various regions of the array, and to share subcomputations of adjacent elements.
- A declarative API that allows us to write cache-friendly programs which access data in a block-wise manner, while cleanly separating the evaluation code from the specification of the array elements.
- A summary of the hairy details of array fusion that one must account for to obtain efficient code, including the staging of inliner phases and the delicate interaction between GHC and its back-end code generator, LLVM.
- Finally, with the ultimate aim of writing declarative code that runs as fast as competing libraries, we discuss the current challenges to array fusion and suggest future directions for research.

The Ypnos [20] and PASTHA [16] libraries also address stencil convolution in Haskell, though [20] presents no performance figures and [16] lacks absolute numbers. On the other hand, Ypnos deals elegantly with arrays of higher rank, and PASTHA also has support for managing convergence conditions for iterative convolution, which we don’t address here. Our philosophy of array programming is also shared by the Chapel language [3], which stresses the need to define the value of each array declaratively, and map this onto physical processors in separate, orthogonal code.

## 2. The Laplace Equation, Reloaded

Although we have found the general principle of Repa’s array representation to work well, when applied to the problem of stencil convolution we now have enough experience with it to point out several infelicities. We will reuse our example from [13] of the numerical solution of the Laplace equation. The overall structure of this example is similar to the code in the original Canny implementation which we are trying to improve.

The `solveLaplace` function in Figure 1 solves the Laplace equation in a 2D grid, with constant value boundary conditions. Numerically, the equation can be solved by applying the following update function to every point in a grid until we reach a fixed point:

$$u'(i, j) = (u(i-1, j) + u(i+1, j) + u(i, j-1) + u(i, j+1)) / 4$$

This has the same effect as convolving the input image with the Laplace stencil shown in Figure 3, and then dividing every element in the result by four. Although in practice we would iterate the above function to a fixed point, for benchmarking we simply iterate it a fixed number of times, hence the `steps` parameter to `solveLaplace`. The boundary conditions are specified by two arrays, `arrBoundValue` and `arrBoundMask`. The first gives the value to use at a particular point, while the second contains 0 where the boundary applies and 1 otherwise. If we are too close to the border of the array to apply the update function, then we return the original value. The `traverse` function used in `relaxLaplace` produces a new array by calling `elemFn` for every index in the result. The `elemFn` worker is passed a lookup function `get`, which it can use to get values from the source array. The type of `traverse` given in the same figure. The expression `(Z :. i :. j)` is an array index to row `i` and column `j`. See [13] for further details.

Although `solveLaplace` gives the correct answer, it has several runtime performance problems:

1. We test for the border at every iteration (the call to `isBorder` in `elemFn`), even though in the vast majority of iterations we are far from it. We will discuss border handling further §4.1.
2. Every lookup of the source array must be bounds checked by the library implementation. Concretely, the user-defined `elemFn` might apply `get` to an out-of-bounds index (if, say, `isBorder` was not implemented correctly), so `get` must conservatively check bounds before indexing the array.
3. Parallelising the potential random access of `elemFn` implies that we must perform indexing computations of the form `x + y*width` for every lookup of the source array, that is, every application of `get`.

We will return to these problems in later sections, but for now note that the bounds checking overhead is the easiest to mitigate, as we can simply disable it. Replacing the use of `(!)` in the definition of `traverse` with an “unsafe” indexing operator removes the overhead, but this is clearly unsatisfying. Far better would be to write the code so that it is correct by construction. Nevertheless, in Figure 2 we present part of the Core IR for the inner loop of an unsafe version of our `solveLaplace` function. This is the code resulting from array fusion, which includes unfolding all of the library functions, inlining the user defined functions into them, and performing a large number of code transformations. The actual loop construct is defined as a part of `force`. We will return to this code in §5. Note that the code that checks for the border is “further up”, so Figure 2 simply loads the surrounding elements from the source array, applies the stencil kernel and boundary conditions, and updates the destination.

In the Core IR, infix operators like `##` and `*#` (one hash) work on integers, while operators like `###` and `/##` (two hashes) work on

```
type DIM2 = Z :: Int :: Int
type Image = Array DIM2 Float

solveLaplace :: Int -> Image -> Image -> Image
solveLaplace steps arrBoundMask arrBoundValue arrInit
  = go steps arrInit
  where go 0 arr = arr
        go n arr
          = let arr' = force
              $ zipWith (+) arrBoundValue
              $ zipWith (*) arrBoundMask
              $ relaxLaplace arr
          in arr' `seq` go (i - 1) arr'

{-# INLINE relaxLaplace #-}
relaxLaplace :: Image -> Image
relaxLaplace arr
  = traverse arr id elemFn
  where _ :: height :: width = extent arr

{-# INLINE elemFn #-}
elemFn get d@(Z :: i :: j)
  = if isBorder i j
    then get d
    else (get (Z :: (i-1) :: j)
          + get (Z :: i :: (j-1))
          + get (Z :: (i+1) :: j)
          + get (Z :: i :: (j+1))) / 4

{-# INLINE isBorder #-}
isBorder i j
  = (i == 0) || (i >= width - 1)
    || (j == 0) || (j >= height - 1)

{-# INLINE traverse #-}          {- LIBRARY CODE -}
traverse :: Array sh a
  -> (sh -> sh') -> ((sh -> a) -> sh' -> b)
  -> Array sh' b

traverse arr newExtent newElem
  = Delayed (newExtent (extent arr)) (newElem (arr !))
```

**Figure 1.** Old implementation of Laplace using indexing

floats<sup>1</sup>. Hashes imply that these operators work on native, unboxed values. There is no overhead due to boxing, unboxing, or laziness, and each unboxed operator essentially corresponds to a single machine operation. The fact that our (unsafe) inner loop is already so “clean” gives us heart that we may reach the proverbial “C-like performance”. Of course, it would be better if this code was fast *and* safe, instead of just fast.

## 3. Delayed Arrays in Repa

In this section we give a quick summary of Repa’s original array representation, which we will improve over in the next. The main features of Repa are:

- *shape polymorphism*: functions can be written that operate on arrays of arbitrary rank.
- *implicit data parallelism*: functions written with Repa can be run in parallel without any extra work by the programmer.
- *array fusion*: we write array functions in a compositional style, using “virtual” intermediate arrays, but the need to actually create the intermediate arrays is eliminated during compilation.

<sup>1</sup> In GHC proper, `###` and `###` actually work on doubles, but we’re using them for floats for clarity.

```

case quotInt# ixLinear width of { iX ->
case remInt# ixLinear width of { iY ->
writeFloatArray# world arrDest ixLinear
  (### (indexFloatArray# arrBV
    (+# arrBV_start (+# (## arrBV_width iY) iX)))
  (### (indexFloatArray# arrBM
    (+# arrBM_start (+# (## arrBM_width iY) iX)))
  (/## (### (### (###
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# (## (-# width 1) iY) iX)))
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# (## width iY) (-# iX 1))))
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# (## (+# width 1) iY) iX)))
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# (## width iY) (+# iX 1))))
    4.0)))
  })

```

**Figure 2.** Old core IR for solveLaplace using unsafe indexing.

In this paper, as we are dealing with stencils of a fixed rank, shape polymorphism is not of particular help so we will not consider it further. What is of interest is parallelism and fusion. Repa achieves this by using the following representation for arrays, which we will extend in §4.1.

```

data Array sh a
  = Manifest (Vector a)
  | Delayed (sh -> a)

```

Our array type is polymorphic over *sh* (shape), which is the type used for the indices, and *a* which is the type of the elements contained. A manifest array is one represented by real data, that is held in flat unboxed array provided by the `Data.Vector` library. A delayed array is represented by an *element function* that takes an array index and produces the corresponding element. Delayed arrays are the key to Repa’s approach to array fusion. For example, the `map` function for arrays is defined as follows:

```

{-# INLINE map #-}
map :: (Shape sh, Elt a, Elt b)
    => (a -> b) -> Array sh a -> Array sh b
map f arr
  = case arr of
    Manifest vec -> Delayed (\ix -> f (vec ! ix))
    Delayed g -> Delayed (f . g)

```

Here, `Shape` is the class of types that can be used as indices, and `Elt` the class of types that can be used as array elements. Both cases of `map` produce a `Delayed` array, and the second corresponds to the following familiar identity:

$$\text{map } f (\text{map } g \text{ } xs) = \text{map } (f \circ g) \text{ } xs$$

Similar traversal functions such as `zipWith` are defined in the same way. We also support reductions such as `sum` and `fold1`, but do *not* support general filtering operations as the resulting array is not necessarily rectangular. Fusion is achieved via the judicious use of `INLINE` pragmas, and the magic of the GHC simplifier. During compilation, the outer structure of functions such as `map` is eliminated, leaving code that applies the worker function directly to each element of the array. Parallelism is introduced by using the `force` function:

```

force :: (Shape sh, Elt a)
    => Array sh a -> Array sh a

```

For `Manifest` arrays, `force` is the identity. For `Delayed` arrays, `force` allocates a fresh mutable `Vector`, and then forks off

$$\begin{array}{ccc}
\textit{Sobel}_X & & \textit{Roberts}_X & & \textit{Kirsch}_W \\
\begin{bmatrix} -1 & 0 & +1 \\ -2 & 0 & +2 \\ -1 & 0 & +1 \end{bmatrix} & & \begin{bmatrix} +1 & 0 \\ 0 & -1 \end{bmatrix} & & \begin{bmatrix} 5 & -3 & -3 \\ 5 & 0 & -3 \\ 5 & -3 & -3 \end{bmatrix} \\
\\
\textit{PeakPoint} & & \textit{HighPass} & & \\
\begin{bmatrix} -1 & -1 & -1 \\ -1 & 8 & -1 \\ -1 & -1 & -1 \end{bmatrix} & & \begin{bmatrix} 0 & 1 & -1 & 1 & 0 \\ 1 & -2 & 4 & -2 & 1 \\ 1 & 4 & -13 & 4 & 1 \\ 1 & -2 & 4 & -2 & 1 \\ 0 & 1 & -1 & 1 & 0 \end{bmatrix} & & \\
\\
\textit{Binomial}_{7X} & & \textit{Laplace} & & \\
\begin{bmatrix} 1 & 6 & 15 & 20 & 15 & 6 & 1 \end{bmatrix} & & \begin{bmatrix} 0 & 1 & 0 \\ 1 & 0 & 1 \\ 0 & 1 & 0 \end{bmatrix} & & 
\end{array}$$

**Figure 3.** Common Convolution Stencils

several concurrent threads. Each thread is responsible for calling the element function for a subset of array indices, and updating the array with the results. Finally, the array is *frozen*, treating it as constant from then on. This freezing operation is a type-cast only, and does not incur any copying overhead. Importantly, although we use destructive update in the implementation of `force`, as this function allocates the resulting vector itself, it is given a pure interface.


In our implementation, we also include `INLINE` pragmas on the definition of `force`. During compilation, GHC creates a fresh unfolding at each use. In most cases we are left with intermediate code consisting of a loop that computes and updates each value of the array directly, without any intermediate function calls, or boxing/unboxing of numeric values.

Finally, note that the programmer is responsible for inserting calls to `force` in the appropriate place in their code. Forcing the array at different points has implications for sharing and data layout, though in practice we have found there are usually only a small number of places where forcing would “make sense”, so the choice presents no difficulty.

## 4. Stencils, Borders and Partitioned Arrays

Several common stencils are shown in Figure 3. For stencil names written with subscripts, the subscript indicates that it is just one member of a closely related family of stencils. For example,  $\textit{Sobel}_X$  differentiates along the X axis only, but rotating it 90 degrees yields  $\textit{Sobel}_Y$  which differentiates along the Y axis. By “rotate” we mean to permute the coefficients of the matrix, so that +1 is in the top-left in this example. The  $\textit{Sobel}_{X,Y}$  stencils are used in Canny edge detection, while  $\textit{Roberts}_X$  and  $\textit{Kirsch}_W$  also perform discrete differentiation. The  $\textit{PeakPoint}$  stencil is used for noise detection,  $\textit{HighPass}$  is a high-pass filter, and  $\textit{Binomial}_{7X}$  is a low-pass filter. The  $\textit{Laplace}$  stencil is used to compute the average of four surrounding pixels, which we discussed in §2. How these stencils are derived is not important to the discussion, but see [19] for a nice introduction to stencil convolution and other image processing algorithms. For the example stencils, we note several features of computational interest, along with exceptions:

1. All coefficients are statically known.
2. Most coefficients are small integers.
3. Many coefficients are zero.
4. All stencils are symmetric.



**Figure 4.** Application of a 3x3 stencil close to the border.

5. All stencils reuse values for the coefficients.
6. Most stencils fit in a 5x5 matrix.
7. Most stencils are square. (ex *Binomial<sub>7X</sub>*)
8. Most stencils have odd dimensions. (ex *Roberts<sub>X</sub>*)

Points 1 & 2 suggest that we can specialise our stencil functions based on the values of the coefficients. For example, multiplication by two can be achieved by addition, and multiplication by one is no-op. This is opposed to, say, writing a general purpose function that reads coefficients from an array, and performs all multiplications explicitly. Points 3, 4 & 5 suggest that there are savings to be had by common sub-expression and dead-code elimination. Point 6 suggests that being able to handle stencils smaller than a certain fixed size would allow us to support most of the common cases. Points 7 & 8 have implications for border handling, which we discuss in the next section.

#### 4.1 Partitioned Arrays

When implementing stencil convolution, an immediate concern is what to do when the stencil “falls off” the edge of the array. For example, Figure 4 shows the application of a 3x3 stencil close to the top left of the array. In the figure, the white squares indicate the *internal* region of the array, where the stencil is entirely within the array. The grey squares indicate the *border*, where part of the stencil falls outside. There are several ways of handling the border case, with two popular options being to return a constant value (like zero) for out-of-bounds elements, or to return the same value as the nearest in-bounds element.

For our Laplace example we use the second option, but note that for an array of size 512x512, only about 0.8% of the pixels are in the border region. For performant code we clearly don’t want to test whether we are in the border region every time we compute an array element. Instead, we represent the *partitioning* of the array into various regions directly. This allows us to define the result array using element functions specialised to each region, and guarantee the one producing the internal elements is not applied in the border region. In effect, partitioning the array allows us to lift the *if*-expression that tests for the border out of the main loop of our program, and have the library code construct the border and internal regions separately. With partitioned arrays, it does not matter if the element function for the border takes a little longer to evaluate than the one for the internal region, as the former is only applied a small number of times. Provided the simpler, internal case is well optimised, we will still get good overall performance.

Our new data types are shown in Figure 5. An *Array* is defined as an extent, and a list of distinct *Regions*. In the rank 2 case the extent will represent the width and height of the array. Each region has a *Range* that defines the set of indices belonging to the region. A *Range* can either be *RangeAll*, which indicates the entire array, or a *RangeRects* which gives a list of rectangles (of arbitrary rank). Given a *RangeRects*, we can determine whether a particular

```
data Array sh a
  = Array { arrayExtent :: sh
           , arrayRegions :: [Region sh a] }

data Region sh a
  = Region { regionRange :: Range sh
            , regionGen   :: Generator sh a }

data Range sh
  = RangeAll
  | RangeRects { rangeMatch :: sh -> Bool
                , rangeRects :: [Rect sh] }

data Rect sh
  = Rect sh sh

data Generator sh a
  = GenManifest { genVector :: Vector a }

  | forall cursor.
    GenCursored { genMake    :: sh -> cursor
                  , genShift  :: sh -> cursor -> cursor
                  , genLoad   :: cursor -> a }
```

**Figure 5.** New Repa Array Types

index is inside this range either by checking whether it falls in any of the *Rects*, or using the predicate *rangeMatch*. This predicate gives the same result as the former, but can use a more efficient implementation than checking each *Rect* individually. In general, for “local” array operations such as indexing a single element, we use the predicate to quickly determine which region the provided index is in. In contrast, *RangeRects* is used when forcing the entire array, and allows us to create a loop specialised to each region.

Each *Region* also has a *Generator* that encodes how the array elements in that region should be computed. As before, generators of *Manifest* arrays are just flat vectors of unboxed values. Delayed arrays are now represented as *cursored* form. The cursored representation allows us to share indexing computations when forcing adjacent array elements, which is discussed further in §5. The regions of a partitioned array must provide *full coverage*, meaning that every array element must be within some region. Regions are permitted to overlap, with the first one in the list taking precedence.

In general, partitioning an array allows us to write element functions specialised to each region. This specialisation can occur on both a per-element and per-region basis. An example of the first is the optimisation of border handling, which we discussed earlier. An example of the second is to use different loop code to evaluate regions of different sizes. For example, when evaluating a region that is short and wide it is best to operate in a row-wise manner, computing an entire row before moving to the next. This helps to recover sharing between horizontally adjacent elements. In contrast, to evaluate a region that is tall but thin it is best to operate column-wise, to exploit sharing between vertically adjacent elements. As the *Region* type provides a direct description of the size of each region, we can perform this specialisation in the library code. The user invokes it automatically with each application of *force*. We discuss per-region specialisation further in §5.3.

#### 4.2 Bounds checking and co-Stencils

Firstly, with respect to bounds checking, we sheepishly admit that the old version of Repa didn’t actually do any. This was mentioned in [13]. As such, it was possible for buggy client programs to crash at runtime. The trouble is that bounds checking each array access adds a substantial overhead, and the comparison and branching constructs involved interfere with fusion. When tried adding it, by having the *Data.Vector* library check each of the indexing operations on the underlying manifest array, this resulted in a 35% slowdown for our Laplace example applied to a 300x300 array.

Ultimately, the problem is that client code written by the user of a library is “untrusted”, meaning that the library must assume it will index out-of-bounds elements. With respect to the code in Figure 1, without a more “heavy weight” technology like dependent types, or some clever analysis, the compiler cannot prove that when the predicate `isBorder` succeeds, the indexing operations in the `else` branch of `elemFn` are guaranteed to be valid. This problem is compounded by the fact that to support shape polymorphism we must check indices of arbitrary rank against the array bounds. Failing that we could check the linear indexing of the underlying manifest vector, but we would still need to manage the mapping between these and the original indices of arbitrary rank.

Our solution to this problem is to invert the relationship between the stencil definition (`elemFn`) and the source array. Instead of having the (untrusted) `elemFn` fetch elements from the source array itself, we instead write the client code to combine source elements fed to it by the (trusted) library. This is similar to the distinction between recursive and co-recursive functions in work on stream fusion [9], where the latter is the “co-stencil” case. Related work on the Ypnos library [20] mentions the co-monadic structure of grid computations, but does not discuss the relationship with bounds checking.

Figure 6 gives the data type that represents stencils, while Figure 7 contains our new implementation of `solveLaplace`. Figure 6 also gives the definition of `makeStencil` which is a utility function defined by our library. The type `Stencil sh a` specifies a stencil function of shape `sh` that operates on arrays of element type `a`. It consists of a size such as `Z: .3: .3` for the 3x3 case, as well as a zero value and accumulator function which define a fold operation over array elements. Figure 7 shows how to define the Laplace stencil. The `iterateBlockwise` repeatedly applies the provided function to an array, forcing it after each iteration. In this and latter code we have elided `INLINE` pragmas, as well as the `Shape` and `ElT` type class constraints to save space. We have also elided the explicit matches against the input arrays `arrBoundMask`, `arrBoundValue` and `arrInit` that require them to be manifest. These are needed in our concrete implementation for performance reasons, but we hope to improve the compiler so they are not required in future. This is discussed further in §7.4.

The lambda abstraction in the definition of `laplace` defines the *coefficient function* for this stencil. This gives the coefficients for each position in the stencil, and has type `(sh -> Maybe a)`. It gives the coefficient at a particular offset from the *focus* of the stencil, or if that coefficient is zero it returns `Nothing` instead. Handling of zeros is discussed further in the next section. As a syntactic convenience, our library also provides some Template Haskell code to make listing the coefficients easier; an example of this is in the `niceLaplace` function.

The operation of computing the sum-of-products of array elements and stencil coefficients is defined by the `Just` case of `makeStencil`. We could have embedded the coefficient function directly in the definition of `Stencil`, but instead define stencils in terms of a more general fold operation. This leaves the door open for other stencil-like operations that are not expressed as a sum-of-products, such as the creation of a histogram of the neighbourhood of each pixel. Returning to the issue of bounds checking, with our new definitions, client code does not need direct access to the source array at all. All of the library functions used in Figure 7 operate on the whole array at a time, and their safety depends on the correctness of the library, instead of the correctness of the client.

Finally, we note that in virtually all related work using imperative languages it is simply assumed that bounds checking is not performed. The focus of recent papers such as [10] and [14] is usually on optimising cache usage, and they presume the existence of correct, heavily optimised straight line code for computing the indi-

```
data Stencil sh a
  = Stencil { stencilSize :: sh
            , stencilZero :: b
            , stencilAcc  :: sh -> a -> a -> a }

makeStencil :: sh -> (sh -> Maybe a) -> Stencil sh a
makeStencil ex getCoeff
  = Stencil ex 0
  $ \ix val acc
    -> case getCoeff ix of
        Nothing  -> acc
        Just coeff -> acc + val * coeff
```

Figure 6. Stencils and Stencil Construction

```
solveLaplace :: Int -> Image -> Image -> Image
solveLaplace steps arrBoundMask arrBoundValue arrInit
  = iterateBlockwise steps arrInit
  $ zipWith (+) arrBoundValue
  . zipWith (*) arrBoundMask
  . map (/ 4) . mapStencil2 (BoundConst 0) laplace

laplace :: Stencil sh a
laplace  = makeStencil (Z :: 3 :: 3)
  $ \ix -> case ix of
      Z :: 0 :: 1 -> Just 1
      Z :: 0 :: -1 -> Just 1
      Z :: 1 :: 0 -> Just 1
      Z :: -1 :: 0 -> Just 1
      -           -> Nothing

niceLaplace :: Stencil sh a
niceLaplace = [stencil2|
               0 1 0
               1 0 1
               0 1 0 |]
```

Figure 7. Stencil Based Laplace Function.


vidual array elements. In contrast, we are trying to produce a (safe!) general purpose functional array library, which also has support for efficient stencil convolution.

### 4.3 Zeros in stencil definitions

Although some coefficients in a stencil are often zero, we don’t want to waste cycles performing the corresponding multiplications, because they do not affect the final sum of products. The simple, neat and *wrong* solution is to allow terms of the form  $0*x$  in the intermediate code, but then add a GHC rewrite rule [21] to implement the obvious identities  $0*x \equiv 0$  and  $x+0 \equiv x$ . Unfortunately, the first one of these is invalid for standard IEEE-754 floating point numbers because the operation  $0 * \infty$  is supposed to produce NaN (Not a Number). Although this hardly matters for image processing, we still don’t want to add this rule as it would apply globally and we risk breaking other code. Instead, we define the coefficient function to return `Nothing` where the stencil does not apply, and use this to skip over the associated term in `makeStencil`. Never the less, in the literature stencils are usually specified using zeros, so we allow them in our Template Haskell version, but eliminate them while desugaring to the coefficient function.

## 5. Sharing and Cursored Arrays

Suppose we apply a 3x3 stencil to a single internal point in an image, and that every coefficient is non-zero. At the least, this would require nine floating point values to be loaded from the source array, and one store to the result. Now, as the computation of a single point in the result does not depend on any others, we can



**Figure 8.** Overlapping support of four adjacent 3x3 stencils

evaluate elements of the result in an arbitrary order. This makes stencil convolution an embarrassingly parallel operation, which gives us much flexibility in the implementation.

However, as we want our convolution to run with good absolute performance on a finite number of processors, it is often better to impose a specific order of evaluation to improve efficiency. Figure 8 shows the evaluation of four horizontally adjacent points. If we were to evaluate each of these points independently, we would need  $4 \times 9 = 36$  loads of the source array, and four writes to the result. However, evaluating all four points in one operation requires only 18 loads, as well as the four stores to the result. There is also the potential to share the evaluation of array indices, and well as multiplications, depending on the form of the stencil.

The potential for sharing indexing computations can be seen in Figure 2 which shows the Core IR for part of the inner loop of our Laplace example. Although this code only computes a single point in the result, note that the second argument to each application of `indexFloatArray#` produces the offset into the array for each point in the stencil. This is performed with the familiar expression  $x + y * \text{width}$ , where  $x$  and  $y$  are the coordinates of the element of interest. However, as the spacial relationship between the elements is fixed, we could instead compute the index of the focus (center) of the stencil, and then get to the others by adding  $+1/-1$  or  $+\text{width}/-\text{width}$ . In the case where we compute four elements of the result in a single operation, the potential savings for index computations are even greater.

Recovering this sort of sharing is a well known problem in compiler optimisation and is the target of the Global Value Numbering (GVN) [2, 23] transformation performed by some compilers. Unfortunately, no current Haskell compiler implements this transform, so we are not home free yet. However, GHC can now compile via LLVM [24], and LLVM *does* implement a GVN pass. Provided we expose enough of the internal indexing computations, the LLVM compiler will do the rest for us. This brings us to cursored arrays.

### 5.1 Cursors arrays

Recall the new Repa array representation from Figure 5. The definition of element generators is repeated below for reference.

```
data Generator sh a
  = GenManifest { genVector :: Vector a }

  | forall cursor.
    GenCursed { genMake      :: sh -> cursor
              , genShift    :: sh -> cursor -> cursor
              , genLoad     :: cursor -> a }
```

A cursor is an abstract representation of an index into the array. The specific form of the cursor is defined by the *producer* of the array, while the consumer must use the provided cursor functions to access elements. As hinted in the previous section, for stencil functions we represent the cursor by a linear index into the array. Given the coordinates of an element, `genMake` computes the linear

index of the corresponding element, the `genShift` function shifts a cursor by an offset, and `genLoad` produces the array element for a given cursor. Using cursors allows us to avoid repeated indexing computations like  $x + y * \text{width}$ , as we can now just compute the linear index of the centre of the stencil, then shift it around to get the other neighbouring elements.

As well as enabling sharing between index computations, cursored arrays strictly subsume our old delayed array representation. To see this, suppose we added an alternative to our `Generator` type that implemented delayed arrays as given in §3

```
data Generator sh a
  = ...
  | GenDelayed { genGetElem :: sh -> a }
```

It turns out this alternative is unnecessary, because we can write functions to convert between the delayed and cursored representations. Given a cursored array, we construct the element function for the delayed version making a cursor then immediately loading from it. Given a delayed array, we construct the cursored one by using the index itself as the cursor. This is possible due to the existential quantification of the `cursor` type.

```
delayedOfCursed :: Generator sh a -> Generator sh a
delayedOfCursed (GenCursed make _ load)
  = GenDelayed (load . make)

cursedOfDelayed :: Generator sh a -> Generator sh a
cursedOfDelayed (GenDelayed getElem)
  = GenCursed id addIndex getElem

addIndex :: Shape sh => sh -> sh -> sh
addIndex = ...
```

To see that cursored arrays also support the delayed array approach to fusion, note that we can implement `map` by composing its parameter with the `load` function of the cursored array. The following code gives the definition of `mapGen` which operates on the generator. The version for arrays is easily defined in terms of it.

```
mapGen :: (a -> b) -> Generator sh a -> Generator sh b
mapGen f gen
  = case arr of
    GenManifest vec
      -> GenCursed id addDim (\ix -> f (vec ! ix))
    GenCursed make shift load
      -> GenCursed make shift (f . load)
```

Finally, note that although we use cursored arrays internally to the library, there is usually no need for client programs to construct them explicitly. In the clients we have written, arrays are usually constructed with higher level utility functions, and combinators such as `map` and `fold` produce the same result independent of the representation of their arguments.

### 5.2 Applying the stencil

Now that we have the definition for cursored arrays, we can see about creating one. Figure 9 gives the definition of `mapStencil2` which takes the definition of a rank 2 stencil, a source array, and produces a cursored result array. The definitions of the rectangles for the border and internal regions have been elided to save space, so have the `inInternal` and `inBorder` predicates, though they are straightforward.

We have also elided the `INLINE` pragmas for the `make`, `shift` and `load` functions. When compiling with GHC we must define these functions as separate bindings and give them `INLINE` pragmas, instead of writing them as lambda abstractions directly in the body of the `let` expression. This ensures that when an array created with `mapStencil2` is finally forced, these definitions are inlined

```

data Boundary a
  = BoundConst a
  | BoundWrap
  | ...

mapStencil2
  :: Boundary a -> Stencil DIM2 a
  -> Array DIM2 a -> Array DIM2 a

mapStencil2 boundary (Stencil sExtent _ _) arr
  = let (Z :: aHeight :: aWidth) = extent arr
      (Z :: sHeight :: sWidth) = sExtent

      rectsInternal   = ...
      rectsBorder     = ...
      inInternal ix   = ...
      inBorder ix     = ...

      make (Z:.y:.x)
        = Cursor (x + y*aWidth)

      shift (Z:.y:.x) (Cursor offset)
        = Cursor (offset + x + y*aWidth)

      loadBorder ix   = case boundary of ...

      loadInner cursor
        = unsafeAppStencil2 stencil arr shift cursor

  in Array (extent arr)
    [ Region (RangeRects inBorder rectsBorder)
      (GenCursored id addIndex loadBorder)

      , Region (RangeRects inInternal rectsInternal)
        (GenCursored make shift loadInner) ]

unsafeAppStencil2
  :: Stencil DIM2 a -> Array DIM2 a
  -> (DIM2 -> Cursor -> Cursor) -- shift cursor
  -> Cursor -> a

unsafeAppStencil2
  stencil@(Stencil sExtent sZero sAcc)
  arr@(Array aExtent [Region RangeAll (GenManifest vec)])
  shift cursor

  | _ :: sHeight :: sWidth <- sExtent
  , sHeight <= 3, sWidth <= 3
  = template3x3 loadFromOffset sZero

  | otherwise = error "stencil too big for this method"

where getData (Cursor index)
  = vec 'unsafeIndex' index

  loadFromOffset oy ox
  = let offset = Z :: oy :: ox
      cur' = shift offset cursor
      in sAcc offset (getData cur')

template3x3 :: (Int -> Int -> a -> a) -> a -> a
template3x3 f sZero
  = f (-1) (-1) $ f (-1) 0 $ f (-1) 1
  $ f 0 (-1) $ f 0 0 $ f 0 1
  $ f 1 (-1) $ f 1 0 $ f 1 1
  $ sZero

```

**Figure 9.** Applying the stencil to an array.

into the unfolding of the `force` function, as well as the element evaluation function `fillCursoredBlock2` which we will discuss in the next section. If we do not do this, then the definitions would *not* be appropriately inlined, and we would suffer a function call overhead for each application. We will return to this delicate point in §7.4.

The values of the border elements depend on the boundary parameter, and two options are shown at the top of the figure. The inner elements are defined via `unsafeAppStencil2`, which produces a function from the cursor value to the element type. Note that this function requires the provided source array to be manifest, so that elements can be extracted directly from the underlying vector using `unsafeIndex`. We use `unsafeIndex` to access the vector because this function performs no bounds checks, so we do not suffer the associated overhead. The safety of these accesses depends on the correctness of our library code, namely the `rectsInternal` list from `mapStencil2`, so that `loadInner` is not applied too close to the border.

Computation of the inner array elements is performed by the `loadFromOffset` and `template3x3` functions. The latter spells out every possible offset from the centre of a 3x3 stencil. We must spell out these offsets “long hand” instead of writing a recursive function to compute the result because we need each application of `f` to be specialised for the provided offset. During compilation, `f` will be bound to a coefficient function like the one defined in `laplace` of Figure 7. In effect, we are using `template3x3` to select every possible coefficient that the coefficient function could have defined. By virtue of `makeStencil` of Figure 6, if the coefficient function returns a valid coefficient for a particular offset then we end up with a term that multiplies that coefficient with data from the source array. If not, then the `Nothing` branch of `makeStencil` comes in to play and the result is unperturbed.

Sadly, the fact that we must spell out every possible offset means that our `unsafeAppStencil2` function is limited to handling stencils of a particular maximum size. In this case we have set the maximum to 3x3, so that it fits on the page. However, the limit is easy to increase and our concrete implementation currently uses 7x7. Limiting the size of the stencil in this way does not affect what coefficients or zero elements can be used, it just requires the entire stencil to fit inside the template. If we had instead written a recursive version of the template function, then GHC would not inline it, killing performance. In general, repeatedly inlining a recursive function may not terminate, leading to divergence at compile time. We can think of several ways of addressing this issue, but all require modification to the compiler, and we defer further discussion to §7.2. If the stencil does not fit inside the template then we fall back to the standard approach of loading the coefficients into a manifest array and iterating directly over that. This gets the job done, but obviously misses out on the benefits of the cursored approach. A follow on effect of spelling out every offset is that this also limits `mapStencil2` to arrays of rank 2. It is straightforward to write versions for other ranks, but we don’t have a way of doing this polymorphically.

Finally, note that `unsafeAppStencil2` defines a function between a cursor and a single array element. The task of actually filling the result array while exposing sharing between adjacent elements is performed by `fillCursoredBlock2`, which we discuss in the next section.

### 5.3 Filling the array, and interaction with LLVM

Using our original `force` function (not shown), but with cursored arrays, produces a loop whose inner fragment consists of the Core IR shown in Figure 10. This version iterates through the linear indices of the result vector, converts each to a rank 2 index (of the form `Z:.y:.x`) and uses the array indexing operator `!` to evalu-

```

case quotInt# ixLinear width of { iX ->
case remInt# ixLinear width of { iY ->
case +# iX (*# iY width) of { ixCenter ->
writeFloatArray# world arrDest ixLinear
  (### (indexFloatArray# arrBV
    (+# arrBV_start (+# (*# arrBV_width iY) iX)))
  (### (indexFloatArray# arrBM
    (+# arrBM_mask (+# (*# arrBM_width iY) iX)))
  (/*# (### (### (###
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# ixCenter width)))
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# ixCenter 1))))
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# ixCenter (-1))))
    (indexFloatArray# arrSrc
      (+# arrSrc_start (+# ixCenter (*# (-1) width))))
    4.0)))
  })

```

**Figure 10.** New core IR for Laplace with index sharing

```

fillCursoredBlock2
:: Elt a => IOVector a      -- vec
-> (DIM2 -> cursor)         -- makeCursor
-> (DIM2 -> cursor -> cursor) -- shiftCursor
-> (cursor -> a) -> Int      -- loadElem, width
-> Int -> Int -> Int -> Int -- x0 y0 x1 y1
-> IO ()

fillCursoredBlock2 !vec !make !shift !load
!width !x0 !y0 !x1 !y1
= fillBlock y0
where
  fillBlock !y
  | y > y1      = return ()
  | otherwise
  = do fillLine4 x0
      fillBlock (y + 1)
  where
    fillLine4 !x
    | x + 4 > x1      = fillLine1 x
    | otherwise
    = do let srcCur0 = make (Z:.y:.x)
        let srcCur1 = shift (Z:.0:.1) srcCur0
        let srcCur2 = shift (Z:.0:.1) srcCur1
        let srcCur3 = shift (Z:.0:.1) srcCur2

        let val0 = load srcCur0
        let val1 = load srcCur1
        let val2 = load srcCur2
        let val3 = load srcCur3

        touch val0 ; touch val1
        touch val2 ; touch val3

        let !dstCur0 = x + y * width
        unsafeWrite vec (dstCur0) val0
        unsafeWrite vec (dstCur0 + 1) val1
        unsafeWrite vec (dstCur0 + 2) val2
        unsafeWrite vec (dstCur0 + 3) val3
        fillLine4 (x + 4)

    fillLine1 !x
    | x > x1      = return ()
    | otherwise
    = do unsafeWrite vec (x + y * imageWidth)
        (getElem $ makeCursor (Z:.y:.x))
        fillLine1 (x + 1)

```

**Figure 11.** Block evaluation function for cursored DIM2 arrays

ate each element of the delayed array. Although the conversion to and from the linear index appears at the top of the Core code, this is successfully eliminated by the LLVM optimiser, so doesn't appear in the object code. Note how each of the elements of the source array are indexed relative to the cursor `ixCenter`.

To recover sharing between adjacent elements we need to evaluate several in one go, which requires a new version of `force`. The inner loop of this new version is defined by `fillCursoredBlock2` in Figure 11, which is also part of the library. This takes a mutable `IOVector`, along with the functions from a cursored array, and uses them to fill a rectangular block in the vector. Parallelism is introduced by having `force` fork off several threads, and having each fill a particular block of array elements. Doing block-wise evaluation also improves cache usage, as the evaluation of each successive row in a block requires source elements loaded during the evaluation of previous rows.

In the definition of `fillCursoredBlock2` we have manually applied the *unroll-and-jam* transformation [7] to evaluate groups of four consecutive elements per iteration. We operate row-wise, which is good for blocks that are at least four elements wide. To evaluate narrow blocks such as the 1 pixel wide left-hand border from Figure 4 it is better to operate column-wise using a separate filling function derived from the presented code.

The `touch` function in the inner loop is used to place a dependency on the computed array values, and prevent GHC from floating the `srcCur*` and `val*` bindings into the applications of `unsafeWrite`. The `touch` function has the following type, and is defined in terms of the GHC primitive operation `touch#` from `GHC.Prim`.

```
touch :: Elt a => a -> IO ()
```

We need all four element values to be computed *before* any of them are written to the result array. This is to avoid a hairy interaction with the LLVM optimiser. Specifically, LLVM does not know that the low level representation of the source and result arrays do not alias, nor does it know that the result array and GHC *stack* do not alias. Any write to the result array or stack is assumed to also modify the source array, which invalidates data held in registers at that point. This in turn breaks the GVN (Global Value Numbering) optimisation which we depend on to recover sharing.

The disassembled x86\_64 object code for the inner part of our loop is given in Figure 12. This is for the *Sobel<sub>x</sub>* stencil shown in Figure 3. Floating point loads are marked with round bullets, while floating point stores are marked with diamonds. There are 18 loads and 4 stores, and comparing this to Figure 8 this is the optimal number for such a 3x3 stencil. However, we still have a slight inefficiency due to aliasing issues. Note the repeated instruction `mov 0x6(rbx),rcx` after each floating point store. The `rbx` register contains a pointer to the stack, and each floating point store invalidates the previously loaded value in `rcx`. Aliasing becomes more of a problem when compiling to architectures with insufficient floating point registers. For example 32bit x86 code can only address 8 of the 16 XMM registers available in 64bit mode. If the LLVM compiler runs out of registers it spills values to the stack, which also invalidates previously loaded values. Fixing this will require more work to GHC's LLVM backend, and/or a type system or analysis that recovers the non-aliasing of heap objects.

## 6. Benchmarks

In this section we discuss the performance of our stencil benchmarks. All measurements were taken on a 2GHz 2xQuadCore Intel E5405 Harpertown server which has 6MB of L2 cache per processor and no hardware threading.



```

9b0: mov    0x2e(rbx), rcx
9b4: mov    0x1e(rbx), rdx
9b8: mov    rdx, rsi
9bb: imul   rcx, rsi
9bf: mov    0x36(rbx), rdi
9c3: lea    0x4(r14,rdi,1), r8
9c8: add    r14, rdi
9cb: lea    0x1(rcx), r9
9cf: imul   rdx, r9
9d3: lea    0x2(r9,rdi,1), r10
9d8: mov    0x6(rbx), r11
9dc: mov    0xe(rbx), r15

• 9e0: movss  0x10(r15,r10,4), xmm7
9e7: lea    (r8,r9,1), r10
• 9eb: movss  0x10(r15,r10,4), xmm8
9f2: subss  xmm7, xmm8
9f7: lea    (r8,rsi,1), r10
• 9fb: movss  0x10(r15,r10,4), xmm9
a02: addss  xmm9, xmm9
a07: addss  xmm8, xmm9
a0c: lea    0x2(rsi,rdi,1), r10
• a11: movss  0x10(r15,r10,4), xmm8
a18: movaps xmm8, xmm10
a1c: mulss  xmm0, xmm10
a21: addss  xmm9, xmm10
a26: dec    rcx
a29: imul   rdx,rcx
a2d: add    rcx,r8

• a30: addss  0x10(r15,r8,4), xmm10
a37: lea    0x1(r9,rdi,1), rdx
• a3c: movss  0x10(r15,rdx,4), xmm9
a43: lea    0x3(r9,rdi,1), rdx
• a48: movss  0x10(r15,rdx,4), xmm11
a4f: subss  xmm9, xmm11
a54: lea    0x3(rsi,rdi,1), rdx
• a59: movss  0x10(r15,rdx,4), xmm12
a60: addss  xmm12, xmm12
a65: addss  xmm11, xmm12
a6a: lea    0x1(rsi,rdi,1), rdx
• a6f: movss  0x10(r15,rdx,4), xmm11
a76: movaps xmm11, xmm13
a7a: mulss  xmm0, xmm13
a7f: addss  xmm12, xmm13
a84: lea    0x3(rcx,rdi,1), rdx
• a89: addss  0x10(r15,rdx,4), xmm13


a90: lea    (rdi,r9,1), rdx
• a94: subss  0x10(r15,rdx,4), xmm7
a9b: addss  xmm8, xmm8
aa0: addss  xmm7, xmm8
aa5: lea    0x1(rcx,rdi,1), rdx
aaa: lea    0x2(rcx,rdi,1), r8
aaf: lea    (rdi,rsi,1), r10
• ab3: movss  0x10(r15,r10,4), xmm7
aba: mulss  xmm0, xmm7
abe: addss  xmm8, xmm7
ac3: movss  0x10(r15,r8,4), xmm8
aca: addss  xmm8, xmm7
acf: lea    (rdi,rcx,1), r8
• ad3: subss  0x10(r15,r8,4), xmm7

ada: add    rax, rdi
add: add    rdi, r9
• ae0: subss  0x10(r15,r9,4), xmm9
ae7: addss  xmm11, xmm11
aec: addss  xmm9, xmm11
af1: lea    (rdi,rsi,1), r8
• af5: movss  0x10(r15,r8,4), xmm9
afc: mulss  xmm0, xmm9
b01: addss  xmm11, xmm9
b06: movss  0x10(r15,rdx,4), xmm11
b0d: addss  xmm11, xmm9
b12: add    rcx, rdi
• b15: subss  0x10(r15,rdi,4), xmm9


b1c: add    r14,rsi
◊ b1f: movss  xmm9,0x10(r11,rsi,4)
b26: mov    0x6(rbx),rcx
◊ b2a: movss  xmm7,0x14(rcx,rsi,4)
b30: subss  xmm11,xmm13
b35: mov    0x6(rbx),rcx
◊ b39: movss  xmm13,0x18(rcx,rsi,4)
b40: subss  xmm8,xmm10
b45: mov    0x6(rbx),rcx
◊ b49: movss  xmm10,0x1c(rcx,rsi,4)
b50: lea    0x8(r14),rcx
b54: lea    0x4(r14),r14
b58: cmp    0x26(rbx),rcx
b5c: jle     9b0

```

**Figure 12.** x86\_64 assembly for *Sobel<sub>X</sub>* applied to four consecutive pixels. FP loads and stores are marked with • and ◊.



**Figure 13.** Laplace Relaxation 300x300 image, 1000 iterations



**Figure 14.** Variation in runtime of Unrolled Stencil Laplace

## 6.1 Laplace again

Figure 13 shows the performance of our Laplace benchmark. The Safe Get version uses the code from Figure 1 with bounds checked indexing, while Unsafe Get is the same code with unchecked indexing. Unsafe Unrolled Get is the bounds checked version, but using the unrolled filling function in Figure 11. Safe Unrolled Stencil uses the unrolled filling function, as well as our cursored arrays. The single threaded Handwritten C version is about 45% faster than our best Haskell result, which is achieved with 3 threads. The C version beats the Haskell one because it doesn't need to do the initial  $x + y * \text{width}$  calculations corresponding to the application of `make` in Figure 11, and there isn't enough sharing inherent in the Laplace stencil for the Haskell version to exploit. For this last point, note the lack of overlap in four Laplace stencils placed side-by-side. Still, it's not an unreasonable result for a Haskell program, considering that the C version produces an inner loop that appears close to optimal. We tried eliminating the application of `make`, but this turned out not to be an improvement due to the extra register required to maintain the centre index between loop iterations.

Figure 13 also contains an important lesson for anyone interested in parallelism in functional languages. The least efficient version of our solver has best speedup graph, yet the most efficient one has the worst. To argue that a particular parallel computing system is useful, one cannot simply present the speedup vs number of cores, as this does not discount the possibility of large linear inefficiencies. In practice we have found that the failure of unboxing or fusion on a given benchmark to cause in excess of a 10x linear slowdown, while maintaining a good speedup graph.

For this benchmark we used an image size of 300x300 matching to our earlier work in [13]. In the end, it appears as though the speedup of this benchmark is limited by scheduling issues. Figure 14 shows the huge variation in runtime for 100 consecutive runs using 4 threads. Increasing the efficiency of our inner loop has also reduced the grain size of the computation. To reduce the variation in runtime for this input size we expect we need *gang scheduling* [11]. Gang scheduling ensures that all threads run in lockstep, instead of being independently scheduled whenever the OS “feels like it”. This is a feature which the GHC runtime does not yet provide.

## 6.2 Sobel Operator

Figure 15 shows the runtimes of the Sobel stencil applied to three image sizes. Also shown is a single threaded version using the `cv::Sobel` function of OpenCV 2.2.0. This is using 32bit floats for the array values. To mitigate variance in runtime due to scheduling issues, we took the best result of 10 runs for each point. In this case, single threaded OpenCV is faster than our single threaded Haskell code primarily because it is using SSE SIMD intrinsics that we do not have access to from Haskell. The LLVM compiler also does not yet support auto-vectorisation to collect separate operations into fused SIMD instructions itself. If this were possible we would not need to use SIMD intrinsics explicitly in our Haskell code. With




Figure 15. Sobel and Canny runtimes, 100 iterations

SSE, the OpenCV version is able to perform loads, stores, additions and multiplications on four packed 32bit floats at a time. However, in all cases we are able to match OpenCV, with the larger image sizes only needing two threads to break even.

### 6.3 Edge Detection

Figure 16 shows the result of applying the Canny algorithm to an example image, with our implementation using two thresholds for edge linking hysteresis. Our implementation is broken into several stages: 1) convert the input RGB image to greyscale; 2) perform a Gaussian blur to suppress high frequency noise; 3) differentiate the image with  $Sobel_{X,Y}$ ; 4) compute magnitude and orientation of the vector gradient; 5) classify local maxima of the gradient into strong and weak edges using the thresholds; 6) select points marked as strong edges; 7) link weak edges that are attached to strong edges. The output consists of all points marked as strong edges, as well as any weak edges that are attached to strong edges. A breakdown of runtimes for each of these stages applied to a 1024x1024 image is shown in Figure 17, while graphs of other sizes are also in Figure 15.

When all is said and done our single threaded implementation is about 4 times slower than OpenCV. With 8 threads it's about 50% slower with a 512x512 image, 10% slower for 768x768, and on par for 1024x1024. We feel this is a good result considering that the blur and differentiation stages for the OpenCV version use SIMD operations that we cannot access from Haskell. The OpenCV implementation also uses different data formats for the various stages, converting between 8-bit unsigned and 16-bit signed integers during the application of  $Sobel_{X,Y}$ . The other stages are




Figure 16. Application of Canny edge detector to an image.

	GCC 4.4.3 OpenCV	GHC 7.0.2 + Repa with # threads			
		1	2	4	8
Grey scale	10.59	12.05	6.19	3.25	2.08
Gaussian blur	3.53	17.42	9.70	5.92	5.15
Detect	18.95	68.73	43.81	31.21	28.49
Differentiate	fused	11.90	7.41	5.38	5.22
Mag / Orient	fused	27.09	16.11	10.45	7.85
Maxima	fused	12.87	7.84	4.83	3.32
Select strong	fused	10.01	5.68	3.60	5.16
Link edges	fused	6.86	6.77	6.95	6.94
TOTAL	33.05	98.25	59.70	40.38	35.72

Figure 17. Canny Edge Detection, 1024x1024 image

performed in a mixture of 8 and 16 bit integer formats. In our own code we also perform the greyscale conversion and edge linking with 8 bit integers. However, using integer operations for the other stages does not help us due to the lack of registers and the aliasing issues mentioned in §5.3.

The OpenCV implementation also hand-fuses the “local maxima” and “select strong” stages, recording an array of indices for strong edges pixels while computing the local maxima. To duplicate this we would need to provide a joint `mapFilter` operation, with a corresponding version of `fillCursoredBlock2`. The delayed array approach cannot recover this form of fusion automatically as it cannot be expressed by simple function composition.

On the positive side, the performance of our Haskell code is more than adequate for real-time edge detection of a video stream. We have an OSX demo available from the Repa homepage [22].

## 7. Challenges of Array Fusion

In this section we summarise the main challenges we have encountered with this work, and suggest avenues for future research.

### 7.1 Lack of support for SIMD operations

At face value, using 4-way SIMD instructions such as available in the SSE or MMX set has the potential to improve the performance of certain algorithms 4-fold. This is assuming that our application isn't memory bound, though note that even a 1024x1024 image of

32bit floats sits comfortably in the 12MB cache of our test machine. Of course, the fact we are using a super-scalar architecture implies that we won't necessarily get a 4-fold speedup on a linear instruction stream, though we note that using SIMD also effectively increases the size of the register set. This would help to avoid the aliasing issues discussed in §5. Whether it's better to introduce SIMD instructions in Haskell code, or have the LLVM compiler reconstruct them is an open question.

## 7.2 Manual unwinding of recursive functions

As mentioned in §5.2 we must manually unfold loops over regions and rectangles as GHC avoids inlining the definitions of recursive functions. The nice way to fix this would be some form of super-compilation [4, 17]. Support for supercompilation in GHC is currently being developed, though still in an early stage. Failing that, we could perhaps add a new form of the `INLINE` pragma that unfolded recursive functions indiscriminately, or a fixed number of times. The downside of the first is potential divergence at compile time, the downside of the second is lack of generality.

## 7.3 Unboxing outside of loops

As mentioned in §4.2 we currently need to add some boilerplate code to functions such as `solveLaplace` to ensure that arrays are unboxed outside of loops, instead of once per iteration. This code has the following form, and is applied to each input array:

```
f arr@(Array _ [RangeAll (GenManifest _)])
  = arr 'deepSeqArray' ...
```

The `deepSeqArray` function places a demand on every boxed object in `arr` before returning its second argument. The pattern match is added to functions that we know will only ever be passed forced arrays, and ensures that indexing operations in the body of the function are specialised for this case. The root problem is that unboxing operations are represented as case matches, but while let-bindings can be floated out of loops, case matches cannot. We hope to fix this particular infelicity in the near future.

## 7.4 INLINES and whole program compilation

As almost every function definition in the Repa library has an `INLINE` pragma, we are essentially doing whole program computation, at least for the array part. In a purely syntactic sense, the `INLINES` do clutter up the code, and we have spent hours hunting performance problems that turned out to be due to a lack of an `INLINE`. In a deeper sense, we feel uneasy about the fact that performance depends so heavily on the syntactic structure of the code, but we don't have a general solution for this. In §5.2 we mentioned the need to write the `make`, `shift` and `load` functions as separate function bindings, and attach `INLINE` pragmas. The need to write separate bindings is simply driven by the need to add `INLINES`, as in GHC this information is attached to the name of the function binding to be inlined.

Although we could imagine adding a desugaring pass that converted the source code to the desired form, in practice we also want to attach inlining stage numbers to many of the bindings, and these must be programmer controlled. Such stage numbers ensure that some bindings are inlined and specialised before others, which can avoid the need for the compiler to optimised large swathes of code only to discard it due to case specialisation later on.

## 7.5 Promises of purity

Figure 18 shows the code for the `force` function that produces a manifest array from a delayed one. This function also has the distinction of being the interface between the IO code that fills the array using concurrent threads, and our pure Repa API. Construct-

```
force :: Array sh a -> Array sh a
force arr
  = unsafePerformIO
    $ do (sh, vec) <- forceIO arr
        return $ sh 'seq' vec 'seq'
            Array sh [Region RangeAll (GenManifest vec)]

where forceIO arr'
      = case arr' of
          Array sh [Region RangeAll (GenManifest vec)]
            -> return (sh, vec)

          Array sh regions
            -> do mvec <- new (size sh)
                mapM_ (fillRegionP mvec sh) regions
                vec <- unsafeFreeze mvec
                return (sh, vec)
```

**Figure 18.** The interface between pure and monadic code.

ing the pure interface consists of two aspects, which are embodied by the following functions:

```
unsafePerformIO :: IO a -> a
unsafeFreeze    :: MVector IO a -> IO (Vector a)
```

The `unsafePerformIO` function breaks the monadic encapsulation of an IO action. Remembering that we're using a lazy language, this is effectively a promise by the programmer that the result can be evaluated at any time without affecting its final value. Similarly, `unsafeFreeze` coerces a mutable vector (`MVector`) to an immutable one (`Vector`), and serves as a promise that after that point in time, the underlying data will not be mutated further. Importantly, failing to respect the two promises results in undefined behaviour at runtime, and neither of the promises can be statically checked by the compiler. Due to this, we would prefer if such promises were enforced by someone else. Actually, the `Data.Vector` library *almost* provides what we want with the following function:

```
create :: (forall s. ST s (MVector (ST s) a)) -> Vector a
```

This function takes an ST action that produces a mutable vector, evaluates it, then coerces the result to an immutable one. The soundness of the coercion is guaranteed by the ST state variable (`s`), which ensures that no references to the mutable vector can escape from the scope of the action that produced it [15]. Unfortunately, there is no equivalent version of `create` for the IO monad, and we need IO because the primitive functions we use to implement parallelism produce IO actions.

More specifically, the `readMVar` and `putMVar` functions operate on mutex variables, and the result of the first can depend on the order in which concurrent threads are scheduled. Note that it is concurrency, not destructive update that is the essential problem here, as destructive update by itself can be safely encapsulated in ST. In related work, the type system of Deterministic Parallel Java (DPJ) [1] can express that concurrent writes to non-overlapping partitions of an array do not interfere. However, DPJ is a first-order language without type inference, and does not support other programming techniques we rely on for Repa. Until such functionality makes it into a higher-order system, it seems that we are forced to use primitives like `unsafePerformIO` and `unsafeFreeze`, and subvert the guarantees of our purely functional language.

On a happy note, although we can't statically check the soundness of our purifying coercions, at least they are confined to a single place in the code – the `force` function. This function is also the logical place for such coercions, as it converts the “abstract” delayed array into a concrete, manifest one.

## 8. Related Work

Facilitating multidimensional array programming and stencil based applications, in particular for parallel target architectures, has been an issue for quite some time. Among others, the Fortran CAPTools toolkit [12], which is designed to automate the process of parallelising sequential Fortran 77 code as much as possible; or Co-array Fortran [18], an extension to Fortran 95 which includes explicit notation for data decomposition.

Ypnos [20] is also a Haskell library which addresses stencil computations. In contrast to Repa, it provides a special stencil pattern matching syntax for one and two dimensional stencils and for historic stencils. The paper does not provide any performance figures, so we cannot compare this aspect to Repa. It would be interesting to investigate whether Repa would be a suitable backend for Ypnos. Another Haskell library, PASTHA [16], whose implementation is based around `IOUArray`. It supports referencing historic stencils, as well as those currently calculated, and includes the specification of the convergence condition as part of the stencil problem definition.

The Accelerate is a high-level embedded language for multidimensional array computations. In contrast to Repa, however, it has at its core an online code generator which targets NVIDIA's CUDA GPGPU programming framework. Accelerate recently gained support for stencil generation as well, but follows a rather different approach, due to its embedded nature.

**Acknowledgements** Thanks to Rami Mukhtar and Ben Lever for writing the original Canny Edge Detection code, and Roman Leshchinskiy for suggesting the use of cursored arrays. This work was supported in part by the Australian Research Council under grant number LP0989507.

## References

- [1] S. V. Adve, S. Heumann, R. Komuravelli, J. Overbey, P. Simmons, H. Sung, and M. Vakilian. A type and effect system for Deterministic Parallel Java. In *In Proc. Intl. Conf. on Object-Oriented Programming, Systems, Languages, and Applications*, 2009.
- [2] B. Alpern, M. N. Wegman, and F. K. Zadeck. Detecting equality of variables in programs. In *Proc. of the 15th Symposium on Principles of Programming Languages*, pages 1–11, 1988.
- [3] R. Barrett, P. Roth, and S. Poole. Finite difference stencils implemented using Chapel. Technical report, Oak Ridge National Laboratory, 2007.
- [4] M. Bolingbroke and S. Peyton Jones. Supercompilation by evaluation. In *Proc. of the third ACM Haskell Symposium*, pages 135–146. ACM, 2010.
- [5] G. Bradski and A. Kaehler. *Learning OpenCV: Computer Vision with the OpenCV Library*. O'Reilly Media, 2008.
- [6] J. Canny. Finding edges and lines in images. Technical report, Massachusetts Institute of Technology, Cambridge, MA, USA, 1983.
- [7] S. Carr, C. Ding, and P. Sweany. Improving software pipelining with unroll-and-jam. In *Proc. of the 29th Hawaii International Conference on System Sciences*. IEEE Computer Society, 1996.
- [8] M. M. Chakravarty, G. Keller, S. Lee, T. L. McDonell, and V. Grover. Accelerating Haskell array codes with multicore GPUs. In *Proc. of the sixth workshop on Declarative Aspects of Multicore Programming*, pages 3–14. ACM, 2011.
- [9] D. Coutts, R. Leshchinskiy, and D. Stewart. Stream fusion: from lists to streams to nothing at all. In *Proc. of the 12th ACM SIGPLAN International Conference on Functional programming*, pages 315–326. ACM, 2007.
- [10] K. Datta, M. Murphy, V. Volkov, S. Williams, J. Carter, L. Oliker, D. Patterson, J. Shalf, and K. Yelick. Stencil computation optimization and auto-tuning on state-of-the-art multicore architectures. In *Proc. of the ACM/IEEE Conference on Supercomputing*, pages 4:1–4:12. IEEE Press, 2008.
- [11] D. G. Feitelson and L. Rudolph. Gang scheduling performance benefits for fine-grain synchronization. *Journal of Parallel and Distributed Computing*, 16:306–318, 1992.
- [12] C. S. Ierotheou, S. P. Johnson, M. Cross, and P. F. Leggett. Computer aided parallelisation tools (capttools) - conceptual overview and performance on the parallelisation of structured mesh codes. *Parallel Comput.*, 22:163–195, February 1996. ISSN 0167-8191. doi: [http://dx.doi.org/10.1016/0167-8191\(95\)00004-6](http://dx.doi.org/10.1016/0167-8191(95)00004-6). URL [http://dx.doi.org/10.1016/0167-8191\(95\)00004-6](http://dx.doi.org/10.1016/0167-8191(95)00004-6).
- [13] G. Keller, M. M. Chakravarty, R. Leshchinskiy, S. Peyton Jones, and B. Lippmeier. Regular, Shape-polymorphic, Parallel Arrays in Haskell. In *Proc. of the 15th ACM SIGPLAN International Conference on Functional Programming*, pages 261–272. ACM, 2010.
- [14] S. Krishnamoorthy, M. Baskaran, U. Bondhugula, J. Ramanujam, A. Rountev, and P. Sadayappan. Effective automatic parallelization of stencil computations. In *Proc. of the 2007 ACM SIGPLAN Conference on Programming Language Design and Implementation*, pages 235–244. ACM, 2007.
- [15] J. Launchbury and S. L. Peyton Jones. Lazy functional state threads. In *Proc. of the ACM SIGPLAN 1994 conference on Programming Language Design and Implementation*, pages 24–35. ACM, 1994.
- [16] M. Lesiak. PASTHA: parallelizing stencil calculations in Haskell. In *Proc. of the 5th ACM SIGPLAN workshop on Declarative Aspects of Multicore Programming*, pages 5–14. ACM, 2010.
- [17] N. Mitchell. Rethinking supercompilation. In *Proceedings of the 15th ACM SIGPLAN International Conference on Functional Programming*, pages 309–320. ACM, 2010.
- [18] R. W. Numrich. The computational energy spectrum of a program as it executes. *The Journal of Supercomputing*, 52(2):119–134, 2010.
- [19] L. O’Gorman, M. J. Sammon, and M. Seul. *Practical Algorithms for Image Analysis*. Cambridge University Press, 2nd edition, 2008.
- [20] D. A. Orchard, M. Bolingbroke, and A. Mycroft. Ypnos: Declarative, Parallel Structured Grid Programming. In *Proc. of the 5th ACM SIGPLAN workshop on Declarative Aspects of Multicore Programming*, pages 15–24. ACM, 2010.
- [21] S. Peyton Jones, A. Tolmach, and T. Hoare. Playing by the rules: Rewriting as a practical optimisation technique in GHC. In *Proc. of the Haskell Workshop*, 2001.
- [22] Repa. The Repa Home Page, Mar. 2011. <http://trac.haskell.org/repa>.
- [23] B. K. Rosen, M. N. Wegman, and F. K. Zadeck. Global value numbers and redundant computations. In *Proc. of the 15th Symposium on Principles of Programming Languages*. ACM, 1988.
- [24] D. A. Terei and M. M. Chakravarty. An LLVM backend for GHC. In *Proc. of the third ACM symposium on Haskell*, pages 109–120. ACM, 2010.